# $Id: PKGBUILD 136112 2011-08-23 03:20:10Z eric $
# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=libsamplerate
pkgver=0.1.8
pkgrel=1
pkgdesc="Secret Rabbit Code - aka Sample Rate Converter for audio"
arch=('i686' 'x86_64')
url="http://www.mega-nerd.com/SRC/index.html"
license=('GPL')
depends=('libsndfile')
options=('!libtool')
source=(http://www.mega-nerd.com/SRC/libsamplerate-${pkgver}.tar.gz)
md5sums=('1c7fb25191b4e6e3628d198a66a84f47')
sha1sums=('e5fe82c4786be2fa33ca6bd4897db4868347fe70')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  ./configure --prefix=/usr
  make
}

check() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make check
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
}
